```
> restart;alias(sigma=sigma(z),H[n]=H[n](z)):
> S5 := z^2*(diff(H[n], z, z))^2-(-n*(-n+alpha+1)+H[n]+(2*n-z+
  beta-1)*(diff(H[n], z)))^2-4*(diff(H[n], z))*(z*(diff(H[n], z))-H
  [n])*(beta+alpha-(diff(H[n], z)));
```

$$S5 := z^2\left(\frac{\partial^2}{\partial z^2}H_n\right)^2-\left(-n\left(-n+\alpha+1\right)+H_n+\left(2n-z+\beta-1\right)\left(\frac{\partial}{\partial z}H_n\right)\right)^2 \quad (1)$$
$$-4\left(\frac{\partial}{\partial z}H_n\right)\left(z\left(\frac{\partial}{\partial z}H_n\right)-H_n\right)\left(\beta+\alpha-\left(\frac{\partial}{\partial z}H_n\right)\right)$$

```
> JMOeq:=(z*diff(sigma,z,z))^2-(2*diff(sigma,z)^2-z*diff(sigma,z)+
  sigma)^2+4*product(diff(sigma,z)+k[j],j=0..3);
```

$$JMOeq := z^2\left(\frac{\partial^2}{\partial z^2}\sigma\right)^2-\left(2\left(\frac{\partial}{\partial z}\sigma\right)^2-z\left(\frac{\partial}{\partial z}\sigma\right)+\sigma\right)^2+4\left(\frac{\partial}{\partial z}\sigma+k_0\right)\left(\frac{\partial}{\partial z}\sigma \quad (2)$$
$$+k_1\right)\left(\frac{\partial}{\partial z}\sigma+k_2\right)\left(\frac{\partial}{\partial z}\sigma+k_3\right)$$

```
> eq1:=collect(expand(subs(H[n]=sigma+b*z+c,S5)-JMOeq),[diff,sigma,
  z],factor):
> solve({4*alpha+2-4*n-8*b+2*beta,2*c+2*n^2+4*b^2-4*alpha*b-2*n*
  alpha-2*beta*b-2*b-2*n+4*n*b},{b,c}):
> H[n]=collect(subs(%,sigma+b*z+c),[z,n],factor):collect(%-sigma,
  [n],factor);
```

$$H_n-\sigma=-\frac{1}{2}n^2+\left(\frac{1}{2}-\frac{1}{2}\beta-\frac{1}{2}z\right)n+\left(\frac{1}{4}\beta+\frac{1}{4}+\frac{1}{2}\alpha\right)z+\frac{1}{8}\left(2\alpha+\beta+1\right)^2 \quad (3)$$

```
> restart;with(PDEtools):with(linalg):with(VectorCalculus):with
  (LinearAlgebra):alias(H[n]=H[n](z),sigma=sigma(z),phi=phi(z));
```

$$H_n, \sigma, \phi \quad (4)$$

```
> S5 := z^2*(diff(H[n], z, z))^2-(-n*(-n+alpha+1)+H[n]+(2*n-z+
  beta-1)*(diff(H[n], z)))^2-4*(diff(H[n], z))*(z*(diff(H[n], z))-H
  [n])*(beta+alpha-(diff(H[n], z)));
```

$$S5 := z^2\left(\frac{\partial^2}{\partial z^2}H_n\right)^2-\left(-n\left(-n+\alpha+1\right)+H_n+\left(2n-z+\beta-1\right)\left(\frac{\partial}{\partial z}H_n\right)\right)^2 \quad (5)$$
$$-4\left(\frac{\partial}{\partial z}H_n\right)\left(z\left(\frac{\partial}{\partial z}H_n\right)-H_n\right)\left(\beta+\alpha-\left(\frac{\partial}{\partial z}H_n\right)\right)$$

```
> K2 := diff(phi, z, z) = (alpha+beta+1)*phi/z-(beta+1-z)*(diff
  (phi, z))/z:K3:=diff(K2,z):K4:=diff(K3,z):
> n:=2;
```

$$n := 2 \quad (6)$$

```
> tau[n]:=collect(subs(K3,K2,det(Wronskian([exp(-z)*phi,seq(diff
  (exp(-z)*phi,z$j),j=1..n-1)],z))),diff(phi,z),factor):
> H[n]:=convert(simplify(subs(K4,K3,K2,z*diff(ln(tau[n]),z))),
  parfrac,diff(phi,z));
```

$$H_2(z) := -\beta-z-1+\frac{\phi\left(-\left(\frac{\partial}{\partial z}\phi\right)\beta-\left(\frac{\partial}{\partial z}\phi\right)+\phi+\phi\beta+\phi\alpha\right)}{\left(\frac{\partial}{\partial z}\phi\right)^2 z+\left(-z\phi+\phi\beta+\phi\right)\left(\frac{\partial}{\partial z}\phi\right)-\phi^2-\phi^2\beta-\phi^2\alpha} \quad (7)$$

```
> subs(K3,K2,S5):collect(%,[diff(phi, z),z,phi],factor);
```

$$0 \quad (8)$$