```
> restart;with(linalg):with(LinearAlgebra):alias(u=u(zeta),phi[nu]=
  phi[nu](zeta)):
> P3:=(diff(u, zeta))/zeta+diff(u, zeta, zeta)-(diff(u, zeta))^2/u-
  (1/2)*alpha*u^2/zeta^2-(1/2)*beta/zeta-Gamma*u^3/zeta^2-delta/u;
```

$$P3 := \frac{\frac{\partial}{\partial \zeta} u}{\zeta} + \frac{\partial^2}{\partial \zeta^2} u - \frac{\left(\frac{\partial}{\partial \zeta} u\right)^2}{u} - \frac{1}{2}\,\frac{\alpha\, u^2}{\zeta^2} - \frac{1}{2}\,\frac{\beta}{\zeta} - \frac{\Gamma\, u^3}{\zeta^2} - \frac{\delta}{u} \tag{1}$$

```
> n:=2;epsilon[1]:=1;epsilon[2]:=1;
```

$$n := 2$$

$$\varepsilon_1 := 1$$

$$\varepsilon_2 := 1 \tag{2}$$

```
> K:=(n)->(ToeplitzMatrix(p,n)):F:=(n)->(ToeplitzMatrix(q,n)):
> U1:=seq(p[d+1]=psi[nu-n+d],d=0..2*n+1):U2:=seq(q[d+1]=psi[nu-n+
  d+1-epsilon[1]],d=0..2*n+1):
> Y:=seq(psi[nu-n+d]=psi(nu-n+d),d=0..2*n):
> psi:=(nu)->(zeta)^(epsilon[1]*nu/2)*(BesselJ(nu,2*sqrt(epsilon[1]
  *epsilon[2]*zeta))):
> subs(U1,K(n+1));subs(U2,F(n));
```

$$\begin{bmatrix} \psi_\nu & \psi_{\nu-1} & \psi_{\nu-2} \\ \psi_{\nu+1} & \psi_\nu & \psi_{\nu-1} \\ \psi_{\nu+2} & \psi_{\nu+1} & \psi_\nu \end{bmatrix}$$

$$\begin{bmatrix} \psi_{\nu-1} & \psi_{\nu-2} \\ \psi_\nu & \psi_{\nu-1} \end{bmatrix} \tag{3}$$

```
> A:=det(subs(U1,Y,K(n+1))):B:=det(subs(U2,Y,F(n))):
> u:=simplify(-epsilon[1]*zeta*diff(ln(A/B),zeta)):
> alpha:=2*(epsilon[1]*n+nu);beta:=2*epsilon[2]*(1-epsilon[1]*nu+n)
  ;
```

$$\alpha := 2\,\nu + 4$$

$$\beta := -2\,\nu + 6 \tag{4}$$

```
> Gamma:=1;delta:=-1;
```

$$\Gamma := 1$$

$$\delta := -1 \tag{5}$$

```
> collect(numer(simplify(P3)),[BesselJ,zeta],factor);
```

$$0 \tag{6}$$