```
> restart;alias(S[n]=S[n](z),sigma=sigma(z)):
> S5:=(z*diff(S[n],z,z))^2-(n*(n+alpha+beta)-S[n]+(alpha+z)*diff(S
  [n],z))^2-4*diff(S[n],z)*(z*diff(S[n],z)-S[n])*(beta-diff(S[n],z)
  );
```

$$S5 := z^2\left(\frac{\partial^2}{\partial z^2}S_n\right)^2-\left(n\left(n+\alpha+\beta\right)-S_n+\left(\alpha+z\right)\left(\frac{\partial}{\partial z}S_n\right)\right)^2-4\left(\frac{\partial}{\partial z}S_n\right)\left(z\left(\frac{\partial}{\partial z}S_n\right)-S_n\right)\left(\beta-\left(\frac{\partial}{\partial z}S_n\right)\right) \tag{1}$$

```
> JMOeq:=(z*diff(sigma,z,z))^2-(2*diff(sigma,z)^2-z*diff(sigma,z)+
  sigma)^2+4*product(diff(sigma,z)+k[j],j=0..3);
```

$$JMOeq := z^2\left(\frac{\partial^2}{\partial z^2}\sigma\right)^2-\left(2\left(\frac{\partial}{\partial z}\sigma\right)^2-z\left(\frac{\partial}{\partial z}\sigma\right)+\sigma\right)^2+4\left(\frac{\partial}{\partial z}\sigma+k_0\right)\left(\frac{\partial}{\partial z}\sigma+k_1\right)\left(\frac{\partial}{\partial z}\sigma+k_2\right)\left(\frac{\partial}{\partial z}\sigma+k_3\right) \tag{2}$$

```
> eq1:=collect(expand(subs(S[n]=sigma+b*z+c,S5)-JMOeq),[diff,sigma,
  z],factor):
> coeff(eq1,diff(sigma,z)):bc:=solve({op(1,%),op(2,%)},{b,c}):
> S[n]=collect(subs(%,sigma+b*z+c),z,factor):collect(%-sigma,[n],
  factor);
```

$$S_n-\sigma=n^2+\left(\alpha+\beta\right)n+\frac{1}{8}\left(2\beta+\alpha\right)\left(\alpha+2z+2\beta\right) \tag{3}$$

```
> eq2:=collect(expand(subs(bc,eq1)),[diff,sigma,z],factor):
> solve({seq(coeff(eq2,diff(sigma,z),j),j=0..3)},{k[0],k[1],k[2],k
  [3]}): %[1];
```

$$\left\{k_0=-\frac{1}{2}\beta-\frac{3}{4}\alpha-n,\ k_1=\frac{1}{2}\beta+\frac{1}{4}\alpha+n,\ k_2=\frac{1}{2}\beta+\frac{1}{4}\alpha,\ k_3=-\frac{1}{2}\beta+\frac{1}{4}\alpha\right\} \tag{4}$$

```
> restart;with(PDEtools):with(linalg):with(VectorCalculus):with
  (LinearAlgebra):alias(S[n]=S[n](t),sigma=sigma(t),phi=phi(t));
```

$$S_n,\ \sigma,\ \phi \tag{5}$$

```
> S5:=(t*diff(S[n],t,t))^2-(n*(n+alpha+beta)-S[n]+(alpha+t)*diff(S
  [n],t))^2-4*diff(S[n],t)*(t*diff(S[n],t)-S[n])*(beta-diff(S[n],t)
  );
```

$$S5 := t^2\left(\frac{\partial^2}{\partial t^2}S_n\right)^2-\left(n\left(n+\alpha+\beta\right)-S_n+\left(\alpha+t\right)\left(\frac{\partial}{\partial t}S_n\right)\right)^2-4\left(\frac{\partial}{\partial t}S_n\right)\left(t\left(\frac{\partial}{\partial t}S_n\right)-S_n\right)\left(\beta-\left(\frac{\partial}{\partial t}S_n\right)\right) \tag{6}$$

```
> K2 := diff(phi, t, t) = (beta+1)*phi/t-(-alpha-2*n+2-t)*(diff
  (phi, t))/t:K3:=diff(K2,t):K4:=diff(K3,t):
> n:=2;
```

$$n := 2 \tag{7}$$

```
> tau[n]:=collect(subs(K3,K2,det(Wronskian([exp(-t)*phi,seq(diff
  (exp(-t)*phi,t$j),j=1..n-1)],t))),diff(phi,t),factor):
> S[n]:=convert(simplify(subs(K4,K3,K2,t*diff(ln(tau[n]),t))),
  parfrac,diff(phi,t));
```

(8)

$$S_2(t) := \alpha + 2 - t + \frac{\phi\left(\left(\frac{\partial}{\partial t}\phi\right)\alpha + 2\left(\frac{\partial}{\partial t}\phi\right) + \phi + \phi\beta\right)}{\left(\frac{\partial}{\partial t}\phi\right)^2 t + \left(-2\phi - t\phi - \alpha\phi\right)\left(\frac{\partial}{\partial t}\phi\right) - \phi^2 - \phi^2\beta} \tag{8}$$

```
> subs(K3,K2,S5):collect(numer(%),[diff(phi, t)],factor);
```

$$0 \tag{9}$$